



Le code de réseau DRAGON5

A. Hébert

2020/06/05

Table des matières

Le code de réseau
DRAGON5
Schéma SN simple
niveau
Schémas REP
double-niveaux
Ressources

Le code de réseau DRAGON5
Schéma SN simple niveau
Schémas REP double-niveaux
Ressources

Le code de réseau DRAGON5

Le code de réseau DRAGON5

Schéma SN simple

niveau

Schémas REP

double-niveaux

Ressources

Le code DRAGON5 couvre différents domaines d'applications

- solveur de l'équation de transport de Boltzmann (BTE) pour les neutrons et les photons
- permet d'étudier une cellule unitaire ou un assemblage de combustible en situation de **mode fondamental**
 - ◆ situation où la géométrie possède des conditions frontières conservatives (réflexion, translation, symmétrie)
- générateur de sections efficaces pour un code de simulation de cœur entier, tel que DONJON5
- permet la programmation de **schémas de calcul** grâce au macro-langage CLE-2000
 - ◆ les schémas de calcul ne sont pas inclus dans la distribution DRAGON5
- La distribution DRAGON5 est offerte en Open Source, sous license GPL
 - ◆ les schémas de calcul peuvent être développés sous license propriétaire.

Le code de réseau DRAGON5


Le code de réseau DRAGON5

Schéma SN simple
niveau

Schémas REP
double-niveaux
Ressources

DRAGON5 est un code de réseau de seconde génération avec les fonctionnalités suivantes:

1. Accès aux bibliothèques de sections efficaces et interpolation en température.
2. Calcul d'autoptotection des résonances.
3. Calcul de flux principal
4. Homogénéisation et condensation des taux de réactions.
5. Calcul des facteurs d'équivalence SPH.
6. Calcul d'évolution isotopique (solution des équations de Bateman).
7. Création d'une base de données réacteur multi-paramètres.



Le code de réseau DRAGON5

Le code de réseau DRAGON5

Schéma SN simple
niveau


Schémas REP
double-niveaux

Ressources

main DRAGON distributions

- **Version3:** La distribution Industrial Standard Toolset de l'industrie nucléaire Canadienne. Maintenue par Guy Marleau.
- **Version4:** La distribution avancée CANDU+PWR. Maintenue par Alain Hébert.
- **Version5:** La distribution 64-bit clean. Maintenue par Alain Hébert.

DRAGON roadmap



Choisir une distribution sur la webpage: <http://merlin.polyml.ca/version5.htm>

- Version5 beta archive. To expand the archive, type "tar xvfz version5_v5.0.1.tgz".

tagged version 5.0.1	tgz	2014/12/17	
tagged version 5.0.2	tgz	2016/02/02	what's new
tagged version 5.0.3	tgz	2017/02/24	what's new
tagged version 5.0.4	tgz	2018/04/22	what's new
tagged version 5.0.5	tgz	2019/01/18	what's new
tagged version 5.0.6	tgz	2020/02/01	what's new

Le code de réseau DRAGON5

Webpage <http://merlin.polymtl.ca/version5.htm>

Le code de réseau DRAGON5

Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

What is the Version5 project

Version5 is a 64-bit clean distribution of the reactor physics codes developed at École Polytechnique de Montréal.[\[1\]](#), [\[2\]](#) Release 5.0.1 is iso-functional with Version4 distribution at release 4.1.0. The Ganlib5 kernel is mostly programmed in ANSI-C language and the computational modules are programmed in Fortran 2003. An introduction of reactor physics is presented in Ref. [\[3\]](#). Other frequently asked questions are answered in our [faq](#) page.

Version5 components

The available components in Version5 distribution are:


- Ganlib5 developer's guide ([PDF](#))
- Dragon5 user's guide ([PDF](#))
- Trivac user's guide ([PDF](#))
- Donjon5 user's guide ([PDF](#))
- Dragon/Trivac object structure guide ([PDF](#))
- CLE-2000 tutorial ([PDF](#))
- NXT module theory guide (geometry numbering) ([PDF](#))
- Version5 beta archive. To expand the archive, type "tar xvfz version5_v5.0.1.tgz".

tagged version 5.0.1	tgz	2014/12/17
tagged version 5.0.2	tgz	2016/02/02
tagged version 5.0.3	tgz	2017/02/24
tagged version 5.0.4	tgz	2018/04/22
tagged version 5.0.5	tgz	2018/12/31

- Open-source Draglibs in XMAS or SHEM binary formats. If you want to download such a multigroup cross-section library in Draglib format, please go [here](#).
- Open-source Wimslib libraries from the [WLUP](#) project. These libraries are distributed in ascii format. To convert them to binary, you have to use the willie utility, available in [for_src](#).
- C++ classes
 - i. [Skin++ classes in Version5](#)
 - ii. [Yacs++ classes in Version5](#)

Development versions

If you need an alpha or beta development version set between two tagged versions, please go [here](#).



Le code de réseau DRAGON5

Le code de réseau

DRAGON5

Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

Installation de DRAGON5

Depuis la version 5.0.5, les makefiles sont disponibles pour faciliter l'installation:

```
tar xvfz version5_v5.0.6.tgz
cd Version5_beta_evx
cd Dragon
make
```

Installation des bibliothèques de sections efficaces


- Télécharger les little-endian draglibs de la page

<http://merlin.polymtl.ca/libraries.htm>

- Copier les bibliothèques dans le répertoire `libraries/l_endian/`

Exécuter le fichier de données

```
cd Version5_beta_evx
cd Dragon
./rdragon tdraglib.x2m
```



Le code de réseau DRAGON5

Le code de réseau
DRAGON5

Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

Modules importants

- Les modules DRAGON5 sont en bleu et les modules TRIVAC5 sont en vert.
- Les modules TRIVAC5 peuvent être utilisés par DRAGON5.
- Les modules DRAGON5 peuvent être utilisés par DONJON5.
- Les codes TRIVAC5, DRAGON5 et DONJON5 sont une collection de modules indépendants, sans connexions entre eux et sans effets de bord.

GEO: Entrée de la géométrie

SYBILT:, SNT:, NXT:, SALT:, etc. Tracking de la géométrie (numérotation, calcul des connexions et des trajectoires pour la méthode SN, PIJ ou MOC

BIVACT:, TRIVAT: Numérotation de la géométrie pour une méthode d'éléments finis dans TRIVAC5

LIB: Fabrication de la microlib (sections efficaces microscopiques par isotope)

SHI:, USS:, TONE: Calcul d'autoprotection des résonances

ASM: Calcul des matrices de PIJ ou des quantités multigroupes non-itératives

BIVACA:, TRIVAA: Calcul des matrices d'éléments finis dans TRIVAC5

FLU: Calcul du flux dans DRAGON5

EDI: Homogénéisation et condensation

SPH: Équivalence SPH

EVO: Solution des équations de Bateman (calcul de burnup)

COMPO: Création d'une multicombo

Le code de réseau DRAGON5

Le code de réseau DRAGON5

Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

Flot de données DRAGON5





Schéma SN simple niveau

Le code de réseau
DRAGON5
Schéma SN simple
niveau
Schémas REP
double-niveaux
Ressources

Ce cas correspond à une géométrie Wigner-Seitz à 7 couronnes avec sections efficaces MATXS à 69 groupes.



```
-----  
* TEST CASE workshop_sn_matxs  
-----  
* Define STRUCTURES and MODULES used  
-----  
LINKED_LIST GEOM TRACK LIBRARY LIBRARY2 ASB FLUX RESEDI ;  
MODULE LIB: GEO: SNT: USS: ASM: FLU: EDI: END: ;  
INTEGER COMBO101 COMBO102 COMBO103 COMBO104 GAIN1  
    MODE1 MODE2 :=  
        1 2 3 4 5 6 7 ; (* DISTRIBUTED SELF-SHIELDING *)  
*  
REAL Rcomb4 := 0.4083 ;  
REAL Rcomb1 := 0.5 SQRT Rcomb4 * ;  
REAL Rcomb2 := 0.8 SQRT Rcomb4 * ;  
REAL Rcomb3 := 0.95 SQRT Rcomb4 * ;  
GEOM := GEO: :: TUBE 7  
    R+ REFL  
    RADIUS 0.0 <<Rcomb1>> <<Rcomb2>> <<Rcomb3>> <<Rcomb4>>  
        0.45 0.5748331 0.6770275  
    MIX 1 2 3 4 5 6 7 ;  
TRACK := SNT: GEOM ::  
EDIT 1 SN 12 SCAT 1 QUAD 1 ;
```

Schéma SN simple niveau

Le code de réseau

DRAGON5

Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

```

-----
* LIB
-----
LIBRARY := LIB: :: EDIT 1
NMIX 7      (*MAXIMUM OF MATERIAL MIXTURES*)
CTRA APOL (*APOLLO TYPE TRANSPORT CORRECTION*)
ANIS 2
SUBG       (*HELIOS TYPE PROBABILITY TABLES*)
*
MIXS LIB: MATXS FIL: MATXS7A
MIX <<COMBO101>> 293.0 (*COMBO101*)
    016      = 016      4.6624E-2   THER 42 FREE
    U235     = U235     7.0803E-4 1  THER 42 FREE
    U238     = U238     2.2604E-2 1  THER 42 FREE
MIX <<COMBO102>> COMB <<COMBO101>> 1.0 (*COMBO102*)
MIX <<COMBO103>> COMB <<COMBO101>> 1.0 (*COMBO103*)
MIX <<COMBO104>> COMB <<COMBO101>> 1.0 (*COMBO104*)
MIX <<GAIN1>> 293.0 (*GAIN1*)
    Zr0      = ZRNAT    4.3241E-2   THER 42 FREE
MIX <<MODE1>> 293.0 (*MODE1*)
    H1       = H1       4.6892E-2   THER 42 H20
    016      = 016      2.3446E-2   THER 42 FREE
MIX <<MODE2>> 293.0 (*MODE2*)
    H1       = H1       4.6892E-2   THER 42 H20
    016      = 016      2.3446E-2   THER 42 FREE
;
-----
* USS
-----
LIBRARY2 := USS: LIBRARY TRACK :: EDIT 1 TRAN PASS 2 ARM ;
-----
* FLUX
-----
ASB := ASM: LIBRARY2 TRACK :: ARM ;
FLUX := FLU: LIBRARY2 TRACK ASB :: EDIT 1 TYPE K ;
RESEDI := EDI: FLUX LIBRARY2 TRACK :: EDIT 5
    MERG MIX
    COND 0.625 ;
ECHO "test workshop_sn_matxs completed" ;
END: ;

```

Schémas REP double-niveaux

Le code de réseau
DRAGON5
Schéma SN simple
niveau
Schémas REP
double-niveaux

Ressources

Travaux réalisé dans le cadre de maîtrises commandités par EDF/R&D et CEA/SPRC. Il s'agit d'un **schéma de calcul** d'assemblage à **deux niveaux**


- Désigné par le nom du dernier étudiant y ayant contribué: **schéma Canbakan**
- Disponibles sur la page <http://merlin.polymtl.ca/canbakan.htm>
- Schéma CLE-2000 programmé avec les modules de DRAGON5
- Pas de modèle de fuites (on utilise TYPE K) ni de contre-réactions (le burnup est le seul paramètre)
- Le premier niveau est basé sur le maillage **SHEM-295**. Le solveur est la **méthode à courants d'interfaces** (IC) avec **itérations flux-courant**
 - ◆ Le calcul d'autoprotection utilise la méthode des sous-groupes avec tables de probabilités CALENDF (méthode SPM, mot clé PT)
- Le second niveau est basé sur un maillage à **26 groupes**. Le solveur est la **méthode des caractéristiques** (MOC)
 - ◆ Discrétisation spatiale de type “moulin à vent” avec description explicite de la lame d'eau
 - ◆ Conditions frontières de réflexion spéculaire (tracking cyclique)
 - ◆ Scattering P_0 corrigé transport ou P_1
 - ◆ Accélération synthétique par la méthode ACA de Igor Suslov
- Le calcul de référence est un calcul Monte-Carlo réalisé avec le code SERPENT2
 - ◆ 2 000 000 neutrons x 1000 cycles à chaque pas
 - ◆ résolution des équations de Bateman par la méthode CRAM

Schémas REP double-niveaux

Le code de réseau
DRAGON5
Schéma SN simple
niveau
Schémas REP
double-niveaux

Ressources

Flot de données



Schémas REP double-niveaux

Le code de réseau
DRAGON5

Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

Détail de l'algorithme en CLE-2000 – cas: UOX+B_TBH_eighth_2level_g2s.x2m

Initialisation des modules, procédures et variables

```

LINKED_LIST GEOSS GEON1 GEON2 TRACKSS TRACKN1 TRACKN2 LIBRARY LIBRARY2
          SYS FLUX FLUX2 EDITION LIBEQ DONNEES LIBHOM BURNUP EDIHOM
          RES EDIOBJ MCOMPO ;
MODULE SYBILT: G2S: SALT: MCCGT: USS: ASM: FLU: EDI: SPH: UTL: DELETE:
          END: EVO: GREP: LIB: COMPO: ;
REAL Rcomb1 Rcomb2 Rcomb3 Rcomb4 R_int_TG R_ext_TG R_int_TI R_ext_TI
          R_int_G R_ext_G Cote CoteL RmodeN1 Lame ;
SEQ_BINARY TF_EXC ;
SEQ_ASCII UOX_TBH :: FILE './UOX_TBH_g2s.txt' ;
SEQ_ASCII BCOND2 :: FILE './UOX_R_BCOND2.txt' ;
SEQ_ASCII FIG_UOX :: FILE './FIG_UOX.ps' ;
SEQ_ASCII _MCOMPO :: FILE './MCOMPO_UOX_TBH.txt' ;
INTEGER an2d := 8 ;
REAL densur := 20.0 ;
INTEGER istep maxstep istepNext maxautop iAutop := 0 73 0 7 1 ;
REAL delr BUbeg BUend Tbeg Tend := 0.01 0.0 0.0 0.0 0.0 ;
REAL Fuelpwr Keff := 36.8 0.0 ;
REAL BU BUautop := 0.0 0.0 ;
PROCEDURE Geo_SS Geo_N1 Geo_N2 Mix_UOX MultLIBEQ ;

```

Initialisation de la Multicompo (un seul paramètre local: **burnup**)

```

MCOMPO := COMPO: :: EDIT 10
STEP UP 'EDI2B'
MAXCAL 5
COMM 'evolution burnup Assembly' ENDC
PARA 'burnup' IRRA
INIT
;

```

Schémas REP double-niveaux

Le code de réseau
DRAGON5
Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

Initialisation des burnups

```

DONNEES := UTL: ::

    CREA
        BURN <<maxstep>> =
            0.0      9.375     18.75      37.5      75.0      150.0
            237.5    325.0     412.5     500.0     625.0     750.0
            1000.0   1250.0    1500.0    1750.0    2000.0    2500.0
            3000.0   3500.0    4000.0    4500.0    5000.0    5500.0
            6000.0   6500.0    7000.0    7500.0    8000.0    8500.0
            9000.0   9500.0    10000.0   10500.0   11000.0   11500.0
            12000.0  12500.0   13000.0   13500.0   14000.0   14500.0
            15000.0  15500.0   16000.0   16500.0   17000.0   17500.0
            18000.0  18500.0   19000.0   19500.0   20000.0   22000.0
            24000.0  26000.0   28000.0   30000.0   32000.0   34000.0
            36000.0  38000.0   40000.0   42000.0   44000.0   46000.0
            48000.0  50000.0   52000.0   54000.0   56000.0   58000.0
            60000.0
;

DONNEES := UTL: DONNEES ::

    CREA
        AUTOP <<maxautop>> =
            4000.0    8000.0    12000.0   24000.0   36000.0   48000.0
            60000.0
;

```

Schémas REP double-niveaux

Le code de réseau
DRAGON5
Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

■ Création de la microlib à 295 groupes

```

STRING Library := "DLIB_295" ;
LIBRARY := LIB: ::

    EDIT 1
    DEPL LIB: DRAGON FIL: <<Library>>
    NMIX 161 CTRA APOL ANIS 2
    PT
    MIXS LIB: DRAGON FIL: <<Library>>
* Moderator
    MIX 1 573.15 NOEV
        H1      = H1_H20   4.857179E-2
        O16     = O16      2.432236E-2
        B10     = B10      4.841415E-6
        B11     = B11      1.948730E-5
* Cladding TI
    MIX 2 573.15 NOEV
        Zr90    = Zr90     2.17736E-2 2 IRSET PT 1
        Zr91    = Zr91     4.74830E-3 2 IRSET PT 1
        Zr92    = Zr92     7.25788E-3 2 IRSET PT 1
        Zr94    = Zr94     7.35521E-3
        Zr96    = Zr96     1.18496E-3
* Fuel -> 8 * 4 evolving mixes
* C1 Cell
    MIX 3 873.15
        U234   = U234     6.9950E-6
        U235   = U235     9.2868E-4 1 IRSET PT 1
        U238   = U238     2.2000E-2 1 IRSET PT 1
        Pu238  = Pu238    1.E-20   1 IRSET PT 1
        Pu239  = Pu239    1.E-20   1 IRSET PT 1
        Pu240  = Pu240    1.E-20   1 IRSET PT 1
        Pu241  = Pu241    1.E-20   1 IRSET PT 1
        Pu242  = Pu242    1.E-20   1 IRSET PT 1
        O16    = O16      4.5871E-2
    MIX 4
        COMB 3 1.0
    MIX 5
        COMB 3 1.0
    MIX 6
        COMB 3 1.0
* cont ...

```

Schémas REP double-niveaux

Le code de réseau
DRAGON5
Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

■ Création des géométries (appel de procédures) et tracking

```

GEOSS := Geo_SS :: <<Rcomb1>> <<Rcomb2>> <<Rcomb3>> <<Rcomb4>>
          <<R_int_TG>> <<R_ext_TG>> <<R_int_TI>>
          <<R_ext_TI>> <<R_int_G>> <<R_ext_G>> <<Cote>>
          <<CoteL>> ;

GEON1 := Geo_N1 :: <<Rcomb1>> <<Rcomb2>> <<Rcomb3>> <<Rcomb4>>
          <<R_int_TG>> <<R_ext_TG>> <<R_int_TI>>
          <<R_ext_TI>> <<R_int_G>> <<R_ext_G>> <<Cote>>
          <<CoteL>> <<RmodeN1>> ;

GEON2 := Geo_N2 :: <<Rcomb1>> <<Rcomb2>> <<Rcomb3>> <<Rcomb4>>
          <<R_int_TG>> <<R_ext_TG>> <<R_int_TI>>
          <<R_ext_TI>> <<R_int_G>> <<R_ext_G>> <<Cote>>
          <<Lame>> ;

UOX_TBH FIG_UOX := G2S: GEON2 ;

TRACKSS := SYBILT: GEOSS ::

  EDIT 0
  MAXR 500
  MAXZ 1000000
  TITLE 'TRACKING FOR ASSEMBLY SS'
  QUA2 20 3
  DP01 ;

TRACKN1 := SYBILT: GEON1 ::

  EDIT 0
  MAXR 500
  MAXZ 1000000
  TITLE 'TRACKING FOR ASSEMBLY N1'
  QUA2 20 3
  DP01 ;

* Level two
TRACKN2 TF_EXC := SALT: UOX_TBH ::

  EDIT 3
  ALLG
  TSPC EQW2 <<an2d>> <<densur>> REND ;

TRACKN2 := MCCGT: TRACKN2 TF_EXC ::

  CACB 4 AAC 80 TMT EPSI 1E-5 MCU 10000 ;

```

Schémas REP double-niveaux

Le code de réseau
DRAGON5
Schéma SN simple
niveau
Schémas REP
double-niveaux
Ressources

■ Calcul d'autoprotection réalisé (méthode des sous-groupes PT) sur le premier niveau

- ◆ Les concentrations isotopiques sont récupérées de LIBRARY2 si `iStep > 1`
- ◆ Le groupement CALC-ENDC permet de regrouper les zones autoprotégées de ces isotopes.

LIBRARY : Microlib de 37 mixtures et 295 groupes d'énergie avec sous-groupes
LIBRARY2: Microlib de 37 mixtures et 295 groupes d'énergie autoprotégée

```

IF iStep 1 = THEN
    LIBRARY2 := USS: LIBRARY TRACKSS :: EDIT 1 PASS 2 ARM GRMIN 52
        CALC REGI W1 U235 ALL REGI W1 Pu238 ALL REGI W1 Pu239 ALL
        REGI W1 Pu240 ALL REGI W1 Pu241 ALL REGI W1 Pu242 ALL
    ENDC ;
ELSE
    GREP: DONNEES :: GETVAL "AUTOP" <<iAutop>> >>BUautop<< ;
    GREP: DONNEES :: GETVAL "BURN" <<iStep>> >>BU<< ;
    IF BUautop BU = THEN
        LIBRARY2 := USS: LIBRARY LIBRARY2 TRACKSS BURNUP :: EDIT 1 PASS 2
            GRMIN 52
            CALC REGI W1 U235 ALL REGI W1 Pu238 ALL REGI W1 Pu239 ALL
            REGI W1 Pu240 ALL REGI W1 Pu241 ALL REGI W1 Pu242 ALL
        ENDC ;
        EVALUATE iAutop := iAutop 1 + ;
    ENDIF ;
ENDIF ;
ENDIF ;

```

■ Calcul de flux du premier niveau par la méthode à courants d'interfaces

- ◆ Utiliser **TYPE K** (pas de fuites) pour les validations versus Monte-Carlo
- ◆ Utiliser **TYPE B** (recherche de buckling critique) pour les calculs de production

```

SYS := ASM: LIBRARY2 TRACKN1 :: EDIT 0 ARM ;
FLUX := FLU: LIBRARY2 SYS TRACKN1 :: 
    EDIT 1 TYPE K ;

```

Schémas REP double-niveaux

Le code de réseau
DRAGON5
Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

- Condensation de la Microlib de 295 à 26 groupes d'énergie et équivalence SPH.
La procédure **MultLIBEQ** permet d'augmenter la Microlib de 37 à 161 mixtures.

LIBEQ: Microlib de 161 mixtures et 26 groupes d'énergie

```
EDITION := EDI: FLUX LIBRARY2 TRACKN1 ::  
    EDIT 0 MICR ALL MERG MIX  
    COND 10 14 18 26 33 40 49 56 66 84 150 210 241 244 247  
    252 255 258 261 268 273 277 281 286 291  
    SAVE ON COND26 ;  
LIBEQ := EDITION :: STEP UP COND26 ;  
LIBEQ := SPH: LIBEQ TRACKN1 :: EDIT 2 ;  
LIBEQ := MultLIBEQ LIBEQ ;
```

- Calcul de flux du second niveau par la méthode des caractéristiques (MOC)

```
SYS := ASM: LIBEQ TRACKN2 TF_EXC :: EDIT 1 ARM ;  
IF istep 1 = THEN  
    FLUX2 := FLU: LIBEQ SYS TRACKN2 TF_EXC ::  
        EDIT 1 TYPE K ;  
ELSE  
    FLUX2 := FLU: FLUX2 LIBEQ SYS TRACKN2 TF_EXC ::  
        EDIT 1 TYPE K ;  
ENDIF ;
```

Schémas REP double-niveaux

Le code de réseau
DRAGON5
Schéma SN simple
niveau
Schémas REP
double-niveaux
Ressources

■ Calcul d'édition à 2 groupes avec isotopes particularisés et fabrication de la Multicompo pour le step courant de burnup

```

EDIOBJ := EDI: FLUX2 LIBEQ TRACKN2 :: EDIT 0
    MICR 35 U234 U235 U236 U237 U238 Np237 Np238 Np239 Pu238 Pu239
    Pu240 Pu241 Pu242 Cm242 Cm243 Cm244 Cm245 Am241 Am242m Am243
    Pm147 Pm148 Pm148m Pm149 Sm147 Sm148 Sm149 Sm150 Nd146 Nd147
    Nd148 B10 B11 Xe135 I135
    MERG COMP
    COND 19
    SAVE ON EDI2B ;

IF istep 1 = THEN
    BURNUP LIBEQ := EVO: LIBEQ FLUX2 TRACKN2 :: EDIT 1 SAVE <<Tbeg>> DAY POWR <<Fuelpwr>>
        SET <<Tbeg>> DAY ;
ELSE
    BURNUP LIBEQ := EVO: BURNUP LIBEQ FLUX2 TRACKN2 :: EDIT 1 SAVE <<Tbeg>> DAY POWR <<Fuelpwr>>
        SET <<Tbeg>> DAY ;
ENDIF ;

MCOMPO := COMPO: MCOMPO EDIOBJ BURNUP LIBRARY :: EDIT 2
    STEP UP 'EDI2B',
    SET <<Tbeg>> DAY ;

EDIOBJ := DELETE: EDIOBJ ;

```

■ Calcul de burnup – solution des équations de Bateman

```

IF istep maxstep < THEN
    EVALUATE istepNext := istep 1 + ;
    GREP: DONNEES :: GETVAL 'BURN' <<istepNext>> >>BUend<< ;
    EVALUATE Tend := BUend Fuelpwr / ;

    BURNUP LIBEQ := EVO: BURNUP LIBEQ FLUX2 TRACKN2 :: EDIT 3 DEPL <<Tbeg>> <<Tend>> DAY POWR <<Fuelpwr>>
        EXPM 1.0E15 GLOB ;

    SYS := DELETE: SYS ;
ENDIF ;

```

Schémas REP double-niveaux

Le code de réseau
DRAGON5
Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

■ Calcul des concentrations isotopiques moyennes dans la Microlib à 37 mixtures

```
* homogenize from 161 mixtures to 37 mixtures and condense to 1-group
* in order to average number densities
EDIHOM := EDI: FLUX2 LIBEQ TRACKN2 :: EDIT 0
    MICR ALL
    COND
    MERG MIX
        1   2   3   4   5   6   7   8   9   10  11  12  3   4   5   6   9
        10  11  12  9   10  11  12  25  3   4   5   6   3   4   5   6   3
        4   5   6   9   10  11  12  9   10  11  12  3   4   5   6   9   10
        11  12  3   4   5   6   9   10  11  12  9   10  11  12  3   4   5
        6   3   4   5   6   3   4   5   6   3   4   5   6   3   4   5   6
        3   4   5   6   9   10  11  12  94  95  96  97  9   10  11  12  9
        10  11  12  94  95  96  97  9   10  11  12  114 115 116 117 114 115
        116 117 122 123 124 125 126 127 128 129 126 127 128 129 126 127 128
        129 126 127 128 129 126 127 128 129 126 127 128 129 126 127 128 129
        154 155 156 157 158 159 160 161
    SAVE ON HOMOGENE
    ;
LIBHOM := EDIHOM :::
    STEP UP HOMOGENE
    ;

* recover number densities from LIBHOM and copy them in LIBRARY2
EDIHOM := DELETE: EDIHOM ;
LIBRARY2 := LIB: LIBRARY2 LIBHOM :: EDIT 0
    MAXS
    MIX   3 MIX   4 MIX   5 MIX   6 MIX   9 MIX  10 MIX  11 MIX  12
    MIX  94 MIX  95 MIX  96 MIX  97 MIX 114 MIX 115 MIX 116 MIX 117
    MIX 122 MIX 123 MIX 124 MIX 125 MIX 126 MIX 127 MIX 128 MIX 129
    MIX 154 MIX 155 MIX 156 MIX 157 MIX 158 MIX 159 MIX 160 MIX 161
    ;
LIBEQ LIBHOM := DELETE: LIBEQ LIBHOM ;
```

■ Sérialisation de la Multicompo

```
_MCOMPO := MCOMPO ;
```

Schémas REP double-niveaux

Le code de réseau
DRAGON5
Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

Principales limitations des schéma de Canbakan

D'importantes fonctions restent à être implémentées dans les schéma de Canbakan pour pouvoir les utiliser à des fins de production industrielle.

- introduire un calcul de fuite (remplacer TYPE K par TYPE B)
- introduire des **calculs de branchement** afin de faire varier certains **paramètres locaux** autour de chaque point de burnup:
 - ◆ température du combustible
 - ◆ température du caloporteur
 - ◆ densité du caloporteur
 - ◆ concentration de bore dans le modérateur
- introduire le calcul des facteurs d'équivalence suivants:
 - ◆ assembly discontinuity factors (ADF)
 - ◆ corner discontinuity factors (CDF)
 - ◆ group form factors (GFF)

Schémas REP double-niveaux

Le code de réseau


DRAGON5

Schéma SN simple
niveau


Schémas REP
double-niveaux

Ressources


Géométries (GD_TBH) utilisées par les schémas REP double-niveaux



Self-shielding



First-level flux



Second-level flux

- Aucune discréétisation du modérateur pour le calcul d'autoprotection
- La géométrie de second niveau utilise une discréétisation **moulin à vent**

Type	Nombre de volumes	Nombre de mixtures	Nombre d'éléments surfaciques
UOX_TBH	801	161	1613
MOX_TBH	801	161	1613
GD_TBH	807	167	1631
UOX_AIC	798	169	1618


Schémas REP double-niveaux

Le code de réseau
DRAGON5
Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

Numérotation des mix combustible des assemblages UOX_TBH et MOX_TBH



Schémas REP double-niveaux

Le code de réseau
DRAGON5
Schéma SN simple
niveau
Schémas REP
double-niveaux
Ressources

Numérotation des mix combustible des assemblages GD_TBH




Figure 3: premier niveau




Figure 4: second niveau

Schémas REP double-niveaux

Le code de réseau
DRAGON5

Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

Numérotation des mix combustible des assemblages UOX_AIC





Figure 5: premier niveau

Figure 6: second niveau

Schémas REP double-niveaux

Le code de réseau
DRAGON5
Schéma SN simple
niveau
Schémas REP
double-niveaux
Ressources

Assemblage UOX_TBH – calcul à deux niveaux



Schémas REP double-niveaux


Le code de réseau
DRAGON5

Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

Distribution des taux de fission – UOX_TBH: burnup=0



Schémas REP double-niveaux

Le code de réseau
DRAGON5
Schéma SN simple
niveau
Schémas REP
double-niveaux

Ressources

Script de récupération des fichiers en sortie – UOX+B_TBH_eighth_2level_g2s.save

```
#!/bin/sh
#
if [ $# = 0 ]
then
echo "usage: UOX+B_TBH_eighth_2level_g2s.save directory" 1>&2
exit 1
fi
echo "access UOX+B_TBH_eighth_2level_g2s.save"
MACH='uname -s'
Sysx="`echo $MACH | cut -b -6`"
if [ $Sysx = "CYGWIN" ]; then
    MACH='uname -o'
elif [ $Sysx = "AIX" ]; then
    MACH='uname -s'
else
    MACH='uname -sm | sed 's/[ ]/_/''
fi
ls -l
mv UOX_R_BCOND2.txt $1/"$MACH"
mv MCOMPO_UOX_TBH.txt $1/"$MACH"
mv UOX_TBH_g2s.txt $1/"$MACH"
mv FIG_UOX.ps $1/"$MACH"
echo "UOX+B_TBH_eighth_2level_g2s.save completed"
```

Schémas REP double-niveaux

Le code de réseau
DRAGON5

Schéma SN simple
niveau

Schémas REP
double-niveaux

Ressources

Exploitation des résultats d'un calcul double-niveaux sur Matlab

- On déplace le fichier `MCOMPO_UOX_TBH.txt` du répertoire `Version5/Dragon/Linux_x86_64/` vers le répertoire `Version5/Octave/data/`.
- On exécute le script suivant:

```
% open the LCM object
FileName = 'MCOMPO_UOX_TBH.txt';
[pfin, ilvl, RecordPos, RecordNumb, RecordName, CurRecInd] = ASCIIOpnv4(FileName);
% step on EDI2B directory
[pfin, ilvl, nbdata, CurRecInd] = ASCIISixv4(pfin, ilvl, RecordPos, RecordNumb, RecordName, ...
CurRecInd, 'EDI2B', 1);
% recover STATE-VECTOR record
ista = ASCIIIGetv4(pfin, ilvl, RecordPos, RecordNumb, RecordName, CurRecInd, 'STATE-VECTOR');
disp(['Main state-vector']); disp([num2str(ista(:))])
% step on DEPL-CHAIN directory
[pfin, ilvl, nbdata, CurRecInd] = ASCIISixv4(pfin, ilvl, RecordPos, RecordNumb, RecordName, ...
CurRecInd, 'DEPL-CHAIN', 1);
% list DEPL-CHAIN directory content
disp(['DEPL-CHAIN directory content'])
ASCIIILibv4(pfin, ilvl, RecordPos, RecordNumb, RecordName, CurRecInd);
% recover ISOTOPESDEPL and DEPLET-DECA records
ista= ASCIIIGetv4(pfin, ilvl, RecordPos, RecordNumb, RecordName, CurRecInd, 'STATE-VECTOR');
nb_iso = ista(1);
hnames = ASCIIIGetv4(pfin, ilvl, RecordPos, RecordNumb, RecordName, CurRecInd, 'ISOTOPESDEPL');
lambda = ASCIIIGetv4(pfin, ilvl, RecordPos, RecordNumb, RecordName, CurRecInd, 'DEPLET-DECA');
% step down to root and close the LCM object
[pfin, ilvl, nbdata, CurRecInd] = ASCIISixv4(pfin, ilvl, RecordPos, RecordNumb, RecordName, ...
CurRecInd, ' ', 0);
fclose(pfin);
% print lambdas
disp([]); disp(['Micro-depleting isotopes lamddas (10^-8 s)'])
for iso=1:nb_iso
    disp([hnames(12*(iso-1)+1:12*iso) ' : ' num2str(lambda(iso))])
end
```

Ressources

Le code de réseau
DRAGON5
Schéma SN simple
niveau
Schémas REP
double-niveaux
Ressources

- **Academic:**
Guy Marleau(guy.marleau@polymtl.ca)
Alain Hébert (alain.hebert@polymtl.ca)
Richard Chambon (richard-pierre.chambon@polymtl.ca)
- **Merlin website:**
DRAGON5/DONJON5:
<http://merlin.polymtl.ca/version5.htm>
- **Archives website:**
Useful informations (including student contributions):
<http://merlin.polymtl.ca/archives.htm>
- **Textbook:**
Alain Hébert, *Applied Reactor Physics*,
Presses Internationales
Polytechnique, Third edition,
Montréal, 2020.

